Asymptotes obliques et termes dominants

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue sous la version Maple 6. …Ça fait un bail ! Dans ce document, les notions d'asymptotes obliques et termes dominants seront présentées par des études de cas.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,display,setoptions):
  setoptions(size=[300,300],axesfont=[times,roman,8],color=navy ):
```

Étude de la fonction f définie par $f(x)=\dfrac{x^2+1}{x}$

Saisissons la fonction f avec l'opérateur flèche.

```
> f:=x->(x^2+1)/x;
```

$$f := x \mapsto \frac{x^2+1}{x} \tag{2.1}$$

Le domaine de la fonction f est $\mathbb{R} \setminus \{0\}$. L'unique candidat à la discontinuité est le nombre 0. Dans le voisinage de $x=0$, le numérateur tend vers 1 et le dénominateur tend vers 0: le quotient est donc de l'ordre de $\pm\infty$. Évaluons donc les deux limites directionnelles en $x=0$.

```
> Limit(f(x),x=0,left)=limit(f(x),x=0,left);
  Limit(f(x),x=0,right)=limit(f(x),x=0,right);
```

$$\lim_{x \to 0-} \frac{x^2+1}{x} = -\infty$$

$$\lim_{x \to 0+} \frac{x^2+1}{x} = \infty \tag{2.2}$$

En $x=0$. il y a alors discontinuité infine et donc, un comportement asymptotique verticale autour de la droite d'équation $x=0$.

Il est clair qu'il n'y a pas d'asymptote horizontale: le degré du numérateur étant supérieur au degré du dénominateur. La fonction f n'a donc aucun comportement asymptotique horizontale: ni dans la partie positive de l'abscisse ni dans la partie négative.

Lorsque le degré du numérateur d'une fonction rationnelle est supérieur de une unité au degré du dénominateur, le tracé de la fonction peut avoir un comportement asymptotique oblique. En effet, la division polynomiale amène toujours un quotient et un reste. Dans le cas où la différence des degré est 1, le quotient est de la forme $ax + b$. Le reste est une cconstante pouvant être, bien sûr, nul. Lorsque le reste est non nul, le tracé de la fonction présente un comportement asymptotique oblique. On dit que la fonction a une asymptote oblique.

Ici, il est facile de voir que l'on a

$$y = \frac{x^2 + 1}{x} \Leftrightarrow y = x + \frac{1}{x}$$

Le quotient et le reste montre que

$$\text{lorsque } x \to \pm\infty,\ y \approx x \text{ et lorsque } x \to 0,\ y \approx \frac{1}{x}$$

– dans le premier cas, nous disons que le terme x domine le terme $\frac{1}{x}$ lorsque x est voisin de $\pm\infty$

– dans le second cas, c'est le terme $\frac{1}{x}$ qui domine lorsque x est voisin de 0

Traçons en couleur khaki les courbes dominantes d'équation $y\_1 = x$ et $y\_2 = \frac{1}{x}$ sur l'intervalle $[-4, 4]$

Contrôlons l'affichage des axes avec l'option `view=[-4..4,-4..4]`.

```
> Courbe_D1:=plot([x,x,x=-4..4],linestyle=3,thickness=0,color=khaki):
  Courbe_D2:=plot([x,1/x,x=-4..4],linestyle=3,thickness=0,
             color=khaki,discont=true):
  display([Courbe_D1,Courbe_D2],scaling=constrained,
                    view=[-4..4,-4..4]);
```



Superposons le tracé de la fonction f. Traçons la fonction f sur l'intervalle $[-4, 4]$ avec une épaisseur moyenne en spécifiant l'option `thickness=2`.

```
> Courbe:=plot([x,f(x),x=-4..4],thickness=2,
                discont=true):
  display([Courbe_D1,Courbe_D2,Courbe],scaling=constrained,
                   view=[-4..4,-4..4]);
```



Puisque, dans le voisinage de $\pm\infty$, le terme dominant est x, la fonction f a donc une asymptote oblique d'équation $y = x$. Observons également qu'au voisinage de $x = 0$, la
courbe semble avoir aussi un comportement asymptotique avec la courbe d'équation $y = \frac{1}{x}$.

La macro-commande `quo` permet d'obtenir le quotient et le reste dans une division polynomiale. Son quatrième argument, optionnel, permet de donner un nom au reste de la division polynomiale. Par commodité, nommez
originalement le reste de la division Reste (il faut que la variable Reste soit une variable libre).

Développons la division polynomiale donnée par la fonction rationnelle f sous la forme d'un quotient plus un reste (division euclidienne).

```
> f(x)=quo(numer(f(x)),denom(f(x)),x,'Reste')+Reste/denom(f(x));
```

$$\frac{x^2+1}{x} = x + \frac{1}{x} \tag{2.3}$$

Résumons: si, pour une fonction rationnelle f définie par $f(x) = \frac{N(x)}{\mathrm{D}(x)}$, le degré de $N(x)$ est d'une unité supérieur au degré de $\mathrm{D}(x)$ et si

$$\frac{N(x)}{\mathrm{D}(x)} = a\,x + b + R(x)$$

où $R(x) \neq 0$, alors la fonction f possède une asymptote oblique d'équation $y = ax + b$.

Étude de la fonction g définie par $g(x) = -\frac{x^3-2}{x^2+1}$

Soit la fonction g définie par $g(x) = -\frac{x^3-2}{x^2+1}$. Le domaine de la fonction rationnelle g est $\mathbb{R}$ car le dénominateur ne s'annule jamais dans $\mathbb{R}$.

Il n'y a donc aucun candidats à la discontinuité. La fonction g n'a donc aucune asymptote verticale.

Puisque le degré du numérateur est supérieur au degré du dénominateur, les limites à moins l'infini et à l'infini n'existe pas. Alors, vous savez d'avance qu'il n'y aura pas d'asymptote
horizontale ni dans la partie négative ni dans la partie positive de l'abscisse.

La différence des degrés des polynômes au numérateur et au dénominateur est de une unité: il est donc possible que la fonction g possède une asymptote oblique. Développons alors la division polynomiale donnée par la fonction rationnelle g sous la forme d'un quotient plus une fraction du diviseur.

```
> g:=x->-(x^3-2)/(x^2+1);
  g(x)=quo(numer(g(x)),denom(g(x)),x,'Reste')+Reste/denom(g(x));
```

$$g := x \mapsto -\frac{x^3-2}{x^2+1}$$

$$-\frac{x^3-2}{x^2+1} = -x + \frac{x+2}{x^2+1} \tag{3.1}$$

Le terme $-x$ domine dans le voisinage de $\pm\infty$. La fonction g possède donc une asymptote oblique d'équation $y = -x$ (le reste est non nul et $R(x) \rightarrow 0$ lorsque $x \rightarrow \pm\infty$.

Superposons l'asymptote oblique et le tracé de la courbe g.

```
> Asymptote_O:=plot([x,-x,x=-4..3],linestyle=3,thickness=0,color=khaki)
  :
  Courbe:=plot([x,g(x),x=-4..3],thickness=2,discont=true):
  display([Asymptote_O,Courbe],scaling=constrained,
              view=[-4..3,-3..4]);
```



Comment peut-on expliquer que dans la partie négative de l'abscisse, la courbe coupe l'asymptote oblique pour ensuite s'en rapprocher indéfiniment ? C'est seulement après
l'introduction aux dérivées et de l'interprétation géométrique de la dérivée seconde que nous serons en mesure de se convaincre de ce comportement asymptotique.

Termes dominants et autres comportement asymptotiques

Étude de la fonction g définie par $f(x) = \dfrac{x^3 - x^2 + x - 2}{x-1}$

Soit la fonction rationnelle f définie par $f(x) = \dfrac{x^3 - x^2 + x - 2}{x-1}$. Le domaine de la fonction f est $\mathbb{R} \setminus \{1\}$. Il y a un seul candidat à la discontinuité: $x = 1$. Évaluons donc les deux limites directionnelles en $x = 1$ (f est toujours définie quelque soit le voisinage de $x = 1$).

```
> f:=x->(x^3-x^2+x-2)/(x-1);
  Limit(f(x),x=1,left)=limit(f(x),x=1,left);
  Limit(f(x),x=1,right)=limit(f(x),x=1,right);
```

$$f := x \mapsto \frac{x^3 - x^2 + x - 2}{x - 1}$$

$$\lim_{x \to 1-} \frac{x^3 - x^2 + x - 2}{x - 1} = \infty$$

$$\lim_{x \to 1+} \frac{x^3 - x^2 + x - 2}{x - 1} = -\infty \quad \textbf{(4.1)}$$

La fonction f possède donc une asymptote verticale d'équation $x = 1$.

Il est évident que la fonction rationnelle f n'a pas d'asymptote horizontale (Pourquoi ?).

De plus, puisque la différence des degrés entre le numérateur et le dénominateur n'est pas de une unité, la fonction n'a pas non plus d'asymptote oblique.

Exprimons quant même la fonction rationnelle sous la forme d'un quotient plus un reste fractionnaire. Cela nous permettra d'analyser les termes dominants.

```
> f(x)=quo(numer(f(x)),denom(f(x)),x,'Reste')+Reste/denom(f(x));
```

$$\frac{x^3 - x^2 + x - 2}{x - 1} = x^2 + 1 - \frac{1}{x - 1} \quad \textbf{(4.2)}$$

On observe donc que

– le terme $x^2 + 1$ domine au voisinage de $\pm\infty$ car $\frac{1}{x-1} \to 0$ lorsque $x \to \pm\infty$

– le terme $\frac{1}{x-1}$ domine au voisinage de 1 car $\frac{1}{x-1} \to \infty$ lorsque $x \to 1$

Superposons les tracés de l'asymptote verticale d'équation $x = 1$ et des deux courbes dominantes d'équation $y\_1 = x^2 + 1$ et $y\_2 = \frac{1}{x-1}$.

```
> Dominant_1:=plot([x,x^2+1,x=-4..4],linestyle=3,thickness=0,color=
  khaki):
  Dominant_2:=plot([x,-1/(x-1),x=-4..4],discont=true,
              linestyle=3,thickness=0,color=khaki):
  Asymptote_V:=plot([1,y,y=-10..10],linestyle=3,thickness=0,color=
  khaki):
  display([Dominant_1,Dominant_2,Asymptote_V],view=[-4..4,-10..10]);
```



Puisque $f(x) = x^2 + 1 - \frac{1}{x-1}$

$$f(x) \approx x^2 + 1 \text{ car } \frac{1}{x-1} \rightarrow 0 \text{ lorsque } \begin{cases} x \rightarrow \infty \\ \text{ou} \\ x \rightarrow -\infty \end{cases}$$

$$f(x) \approx -\frac{1}{x-1} \text{ car } x^2 + 1 \rightarrow 2 \text{ lorsque } x \rightarrow 1$$

Superposons le tracé de la fonction f aux tracés précédents.

```
> Courbe:=plot([x,f(x),x=-4..4],discont=true,thickness=2):
  display([Dominant_1,Dominant_2,Asymptote_V,Courbe],view=[-4..4,-10..10]);
```



On voit que f se comporte à peu près comme la parabole d'équation $y = x^2 + 1$ lorsque $x \rightarrow \infty$ ou lorsque $x \rightarrow -\infty$.

On voit aussi que f se comporte à peu près comme la courbe d'équation $y = \frac{1}{x-1}$ lorsque $x \rightarrow 1$.